


(A)



(B)

```

PREFIX np: <http://www.nanopub.org/nschema#>
PREFIX rso: <http://rdf.biosemantics.org/ontologies/referencesequences#>
PREFIX so: <http://purl.org/obo/owl/SO#>
PREFIX ff: <http://purl.obolibrary.org/obo/>
PREFIX entGene: <http://linkedlifedata.com/resource/entrezgene/>
PREFIX lldRel: <http://linkedlifedata.com/resource/relationontology/>
PREFIX uniprot: <http://purl.uniprot.org/taxonomy/>

select ?cageStart ?cageEnd ?cageChromosome ?cellTypeLabel ?tpmValue where {

# FANTOM5 Ontology
SERVICE <http://agraph.biosemantics.org/catalogs/rajatest/repositories/fantom5-ontology> {
  # CL_0000188 = skeletal muscle cell
  ?cellType rdfs:subClassOf*ff:CL_0000188.
}

# Linked Life Data
{
  ?geneResource entGene:geneSymbol|entrezgene:synonym "MYOD";
  lldRel:expressedInOrganism uniprot:9606 ;
  entGene:hasGeneID ?geneID .
}

# Convert ?geneID to Bio2RDF gene resource URI
BIND (URI(CONCAT("http://bio2rdf.org/geneid:",?geneID)) AS ?bio2rdfGeneID)

# FANTOM5 Nanopublications
SERVICE <http://fantom5.nanopub.org/fantom5/sparql> {

graph ?rikenAssertion2 { # Type 2 Nanopublicatons
  ?cageCluster rso:is_observation_of ?tss.
  ?tss so:part_of ?bio2rdfGeneID.
}

graph ?rikenAssertion1 { # Type 1 Nanopublicatons
  ?cageCluster rso:mapsTo ?cageClusterRegion.
  ?cageClusterRegion rso:start ?cageStart;
  rso:end ?cageEnd;
  rso:regionOf ?cageChromosome.
}

graph ?rikenAssertion3 { # Type 3 Nanopublicatons
  ?cageCluster so:associated_with ?measurementValues;
  rso:observed_in ?cellType.
  ?measurementValues ff:IAO_0000004 ?tpmValue.

  FILTER (?tpmValue >= 0)
}
}
} order by desc (?cageStart)

```